



BSV 101: DESIGNING A COUNTER Using the Bluespec Development Workstation

© Copyright 2005–2009 Bluespec, Inc. All Rights Reserved.

October 27, 2009

1 Introduction

This tutorial aims to explore the basics of design in Bluespec SystemVerilog™ (BSV) using the Bluespec development workstation. We assume previous hardware design experience in Verilog or VHDL and hope that after you've completed this tutorial you will be able to design, synthesize, and debug simple circuits and testbenches in BSV using the development workstation.

2 Starting the Development Workstation

The Bluespec development workstation is a full-featured graphical environment for working with Bluespec designs. The first step, is to start up the workstation by executing the command `bluespec` from a unix command window:

```
bluespec
```

This will bring up Main workstation window from which you can perform any Bluespec tasks.

The basic unit of work within the workstation is the **Project**. A project is a collection of options and settings, saved in a project file (*projectname.bsproj*). Our first task is to create a *new* project called Counter. Select **Project**→**New** (as shown in Figure 1) or the **New** icon on the toolbar (shown in Figure 2).




Figure 1: Project Menu

This will bring up the **New Project** window, as shown in Figure 3. Enter a project directory and a project name. We're going to create a simple counter, so let's name our project **Counter**, as shown in Figure 3.




Figure 2: Toolbar




Figure 3: Create New Project

3 Counter

Now that we have created a project, we are ready to create the design files. Note that when we created the project, the **Project Files** window, shown in Figure 4, opened. This window displays all the files in the project. As you can see in the figure, there are no files in our project yet.

Let's create a file `MyCounter.bsv` to describe our counter. There is a library package called `Counter`, so we're going to use a different name for our package. In the **Project Files** window, select **File**→**New**. Select a directory (where `.` is the current directory) and specify a file name (`MyCounter.bsv`), as shown in Figure 5. This will open an editor window. ¹

First, we need to know how the counter interacts with the outside world. To start with, the user will be able to read the value, load a custom value and increment the counter. In BSV parlance, this means we must define an interface with a method for each action:

```
interface Counter;
  method Bit#(8) read();
  method Action load(Bit#(8) newval);
  method Action increment();
endinterface
```

¹To specify the editor (emacs or vim), use the **Project**→**Options** Files tab.



Figure 4: Project Files Window

(we'll make an eight-bit counter first, and later generalize it to any bit width).

Our `Counter`² interface contains two kinds of methods: a *value* method (`read()`) and two *action* methods (`increment()` and `load()`). Only action methods, distinguished by a return type of `Action`, may modify state (such as register contents) in a module; the typechecker ensures that side effects are not permitted in value methods. Both kinds of methods may take arguments, as `load()` does above.

Now, we need a module, perhaps called `mkCounter()`, to implement (or provide) the `Counter` interface:

```
(* synthesize *)
module mkCounter(Counter);
    // module body goes here
endmodule
```

The `(* synthesize *)` attribute directs `bsc` to generate a separate Verilog module for `mkCounter()`—otherwise, `mkCounter()` will be inlined in the instantiating module.³

Our counter will need to store its current value in a register, which we instantiate like any other module instance:

```
Reg#(Bit#(8)) value <- mkReg(0);
```

²in BSV, interface and type names (such as `Counter` or `Bit` above) begin with a capital letter, whereas module, variable, function, and method names (such as `read()` or `newval`) start with a lowercase letter.

³in general, we prefer to let `bsc` inline instantiated modules; we employ `(* synthesize *)` when the blocks must be separately handled by downstream tools (e.g., RTL synthesis).



Figure 5: Create New File

(the 0 argument to `mkReg` is the reset value of the register).⁴

State by itself, of course, is rather useless: we must allow the user to change and read it via the methods of our `Counter` interface. The `read()` method is the easiest: it just relays the contents of `value`:

```
method Bit#(8) read();
    return value;
endmethod
```

Now, `load()` overwrites `value` with its argument,

```
method Action load(Bit#(8) newval);
    value <= newval;
endmethod
```

and `increment()` adds one to `value` and writes it back:

```
method Action increment();
    value <= value + 1;
endmethod
```

Putting the register instance and the methods together inside the `module ... endmodule` block, we obtain the complete counter module. Save the file.

⁴`mkReg()` creates a register with synchronous reset. An asynchronously reset register can be made with `mkRegA()`, and a register without reset with `mkRegU()`.

We can now synthesize our BSV into Verilog. First we have to check that we have our project configured correctly, through the **Project Options** menu. From the main project window, select **Project**→**Options**. First let's check the compiler options. On the tab marked **Compile**, verify that the compiler target (the `compile to` field) is set to Verilog, as shown in Figure 6. This tab is also where you can add any compiler flags, in the **Options** field. Press **OK** to save your changes and close the form.




Figure 6: Project Options - Compile Tab

We also have to specify the top file in our project. This is done on the **Project**→**Options File** tab. Using the **Browse** button you can select the file we've just created.

To compile, select compile from the tool bar, the Main menu bar, or from within the **Project Files** window. All messages generated by the compiler and the workstation will be displayed in the Main project window,

Note that the compiler (`bsc`) only recompiles files which have changed, so re-running the `bsc` command above will not cause `mkCounter.v` to be regenerated. We can force recompilation by removing the relevant `.bi` and `.bo` files (intermediate files created by `bsc`) or touching the source file. The **Build**→**Clean** task will remove these intermediate files.

4 Testbench

Now, how do we test this counter? Out of the many possible approaches, we shall focus on a simple finite state machine which performs predefined actions at certain times and checks that the counter output matches expectations.

First, we'll write the testbench module skeleton in a file `TbCounter.bsv` (Remember, you can create a new file from the **Project Files** window).

```
import MyCounter::*;

(* synthesize *)
module mkTbCounter();

endmodule
```

You may be surprised that, unlike `mkCounter()`, `mkTbCounter()` leaves out the name of the interface it implements. Because our testbench doesn't need to interact with any external modules, its interface contains no methods; `bsc` assumes such an interface when a module declaration omits the interface type.

Our testbench will instantiate a counter to test and a state register (containing, say, sixteen bits) to keep track of which phase it's in:

```
Counter counter <- mkCounter();
Reg#(Bit#(16)) state <- mkReg(0);
```

Now all we need to do is to add behavior triggered by certain values of `state`. In `mkCounter()`, all behavior was contained in methods and carried out when the methods were invoked by the instantiating module, but `mkTbCounter()` doesn't have any methods! For this purpose, we employ rules—self-triggered methods of sorts:

```
rule step0(state == 0);
    counter.load(42);
    state <= 1;
endrule

rule step1(state == 1);
    if (counter.read() != 42) $display("FAIL: counter.load(42)");
    state <= 2;
endrule
```

We'll also add a rule to end the simulation at a specific time:

```
rule done(state == 2);
    $display("TESTS FINISHED");
    $finish(0);
endrule
```

which completes our testbench. Save the file. We can now synthesize everything to RTL. To synthesize the complete design we need to change the top file and top module (**Project**→**Top File** menu) to `TbCounter.bsv` and `mkTbCounter`. Then compile, from either the **Build** menu or the task bar. The compile will produce the files `mkCounter.v` and `mkTbCounter.v`.

5 Simulation

We can employ Bluespec’s native simulator Bluesim to simulate the design and generate a VCD. To do this, we compile the sources to Bluesim objects instead of Verilog RTL. Open the **Project Options** menu and on the **Compile** tab, change the **Compile to** check box to Bluesim. You’ll have to remove the generated files, using the **Clean** task, so the files will be recompiled. Now select **Compile** from either the **Build** menu or the task bar.

The compiled hardware description has to be linked into the simulation environment before we can simulate the project. The result of the linking stage is a binary which, when executed, simulates a module. In the workstation you set the simulation target in the **Project**→**Options** menu. If the simulation target is Bluesim, you just need to select Bluesim in the **Compile** tab. If the simulation target is a Verilog simulator, you select Verilog on the **Compile** tab, then select the particular Verilog simulator on the **Link/Simulate** tab. If you have any compiler flags you want to specify in the link stage, use the **Options** field on the **Link** tab.

To link, select **Link** from either the **Build** menu or the task bar.

To then run the Bluesim simulator, select **Simulate** from either the **Build** menu or the task bar.

As mentioned above, if you’ve already compiled to Verilog, you have to clean the intermediate files before compiling to Bluesim. The **Full Rebuild** task, available from either the **Build** menu or the task bar performs the following tasks as a single task:

- Full Clean
- Compile
- Link
- Simulate

Alternately, we can simulate the generated Verilog RTL using a Verilog simulator. Because clock and reset signals are by default implicit in BSV, we must use a wrapper which instantiates the module to be simulated and flips the clock every so often. Conveniently, the file `main.v` included in the BSV distribution does just that: it instantiates a module identified by preprocessor macro `TOP` and toggles the clock

every five simulation cycles. `main.v` also dumps all signals to a VCD file whenever it is passed the `+bscvcd` flag. All we have to do is define `TOP` to be `mkTbCounter`, which is done by setting the `top module` field, and compile `main.v` along with the generated RTL.

To simulate Verilog in the workstation, define the following:

- Compile to Verilog (Options, Compile tab)
- Simulator - set to whichever simulator you want to use (Options, Link/Simulate tab)
- Add the `+bscvcd` flag to the simulator (Options, Link/Simulate tab, Simulate options field)
- Top file - `TbCounter.bsv` in our example (Top File)
- Top module - `mkTbCounter` in our example (Top File)

Once you've set your options, you run the following tasks:

- Compile
- Link
- Simulate

or use the **Full Rebuild** task.

Once the simulation is complete, we should see “TESTS FINISHED” as the output in the main window, and obtain a waveform dump in `dump.vcd`. Let us examine the VCD. Follow these steps to view waveforms from the development workstation:

1. Open the **Module Browser** from the **Window** menu.
2. Load the top module (**Module**→**Load Top Module**), as shown in Figure 7. to obtain the module hierarchy from the `.bi/.bo` files.
3. Start or Attach the waveform viewer (**Wave Viewer**→**Start** or **Wave Viewer**→**Attach**) to initiate communication between the workstation and the waveform viewer. The Waveform viewer menu is shown in Figure 8.
4. Load the waveform dump file either from the workstation (**Wave Viewer**→**Load Dump File**) or from within the waveform viewer itself.
5. Select a module or signals and send (**Send to Wave**) to the viewer.




Figure 7: Module Browser - Module Menu




Figure 8: Module Browser - Waveform Viewer Menu

You can modify the waveform viewer settings directly from **WaveViewer**→**Options**.

You’ll notice that the VCD has several signals which do not precisely correspond to registers or wires in the BSV source. The signals named `WILL_FIRE_RL_rulename` are asserted when the relevant rule is being executed, and make it easy to tell which operations our testbench is performing during each clock cycle. Special signals are also generated for each method: in addition to the outputs of *value* methods (e.g., `read`) and method arguments (e.g., `load_newval`), we see a “ready” signal for each method (e.g., `RDY_increment`), and an “enable” signal for each *action* method (e.g., `EN_increment`). Just as the rules’ `WILL_FIRE` signals, these tell us when a given method is ready to be called⁵ and (for action methods) actually invoked, and are

⁵in our `mkCounter()`, all methods are always ready. While this need not always be the case—a method may wait for a computation to complete, for example—in cases where ready signals are always high, the ports can be removed by adding “`always_ready`” to the module’s `(* synthesize`




Figure 9: Waveform Viewer

invaluable in debugging.

6 Decrementing

Let us now extend our counter by adding a `decrement()` method. From the **Project Files** window, open the `MyCounter.bsv` file either by double clicking on the file name, or selecting the file and choosing `edit`. We'll have to extend our interface by adding a declaration for `decrement()` to the `interface ... endinterface` block:

```
method Action decrement();
```

and implement the method in the module block:

```
method Action decrement();
    value <= value - 1;
endmethod
```

That's all! Note that we don't need to modify the logic in any of the other methods—`bsc` will analyze any concurrency conflicts we may have introduced and automatically

*) attribute (`bsc` will report an error unless it can prove that the method is in fact always ready). There is a dual "always_enabled" attribute which removes the enable signals and ensures that the method is always called.

insert interlocks, so each rule and method is guaranteed to do exactly what it says regardless of what other rules and methods exist. What's more, the existing tests in the testbench do not need to be modified to account for the new addition.

While we're on the subject of concurrency, we might well ask what happens when `increment()` and `decrement()` are invoked at the same time (we'd *like* them to cancel each other out). Let us experiment: open the `TbCounter.bsv` file and let's change rule `step1` to call both methods:

```
rule step1(state == 1);
    counter.increment();
    counter.decrement();
    state <= 2;
endrule
```

and recompile. But now `bsc` reports an error:

```
"TbCounter.bsv", line 13, column 10: (G0004) Error:
Rule RL_step1 uses write methods that conflict in parallel:
    counter.increment
and
    counter.decrement
```

It turns out that `bsc` does not know much about algebra, and does not permit `increment()` and `decrement()` to be called concurrently—but putting both calls in the same rule requests just that.

If we can't put the calls in one rule, can we put them in two different rules? Let's try:

```
rule step1a(state == 1);
    counter.increment();
endrule
rule step1b(state == 1);
    counter.decrement();
endrule
rule step1c(state == 1);
    state <= 2;
endrule
```

Now, compilation succeeds, but with two warnings:

```
"TbCounter1.bsv", line 4, column 8: (G0010) Warning:
Rule "step1b" was treated as more urgent than "step1a". Conflicts:
    "step1b" vs. "step1a":
        calls to
```

```

counter.decrement vs. counter.increment
  "step1a" vs. "step1b":
    calls to
counter.increment vs. counter.decrement
"TbCounter1.bsv", line 13, column 14: (G0021) Warning:
  According to the generated schedule, rule "step1a" can never fire.

```

Now, because `increment()` and `decrement()` could not be called in the same clock cycle, the scheduler chose to fire rule `step1b` whenever `state == 1`, thus ruling out `step1a`.

But let us return to the goal of making `increment()` and `decrement()` safe to call concurrently. Implementing our desired behavior will require some more BSV machinery. Conceptually, we would like to specify special behavior when both methods are called. But, because methods cannot communicate among themselves except through instantiated modules, this is not possible directly.

Instead, we'll use a predefined element called `PulseWire`. A `PulseWire` can be set via its `send()` method and read by mentioning its name; because it contains no clocked state, the pulse vanishes at the next clock edge. We'll need a `PulseWire` for each method, which we'll set whenever that method is called; we will then carry out the arithmetic in rules which read the `PulseWires`. In the `mkCounter` module ... `endmodule` block, we instantiate the pulse wires:

```

PulseWire increment_called <- mkPulseWire();
PulseWire decrement_called <- mkPulseWire();

```

and change each method to do nothing but send a pulse down the relevant wire:

```

method Action increment();
  increment_called.send();
endmethod

method Action decrement();
  decrement_called.send();
endmethod

```

All that remains is to implement rules which increment and decrement value whenever one of the `PulseWires` is set:

```

rule do_increment(increment_called && !decrement_called);
  value <= value + 1;
endrule

rule do_decrement(!increment_called && decrement_called);
  value <= value - 1;
endrule

```

Since the value does not change when `increment()` and `decrement()` are invoked simultaneously, no rule is required to cover that case.

7 Better Testbench

Adding tests to our testbench is a bit of a pain: each test requires two new rules, triggered on a unique value of the “current state,” which wiggle the device under test, update the state, check that the device has responded correctly, and update the state again, as well as a change to the `done` rule to match the new “last” state. Fortunately, BSV has a library for describing finite state machines more concisely, which allows us to write a shorter and more maintainable testbench.

Briefly, the FSM library works by capturing a sequence of actions and instantiating a module which automatically builds the FSM to run this sequence one step per clock cycle. The sequence of `actions` is defined using a `seq ... endseq` block and has type `Stmt`. An FSM is then created by passing the sequence to `mkAutoFSM()`.

To add this to our testbench, we need to import the FSM library in addition to our device under test:

```
import StmtFSM::*;
import MyCounter::*;
```

We then define a statement sequence `test_seq` to be executed by the FSM—perhaps set the counter to a value and check that it returns that value in the next cycle:

```
Stmt test_seq = seq
    counter.load(42);
    if (counter.read() != 42) $display("FAIL: counter != 42");
    $display("TESTS FINISHED");
endseq;
```

and instantiate an FSM to sequentially execute the statements in the `seq ... endseq` block, one per clock cycle:⁶

```
mkAutoFSM(test_seq);
```

Under the covers, the `mkAutoFSM()` instantiation creates a new module with an automatically sized state counter, adds a rule for every `action ... endaction` block, much like we did by hand in the original testbench; the resulting FSM runs through the sequence once and exits the simulation.

Our new testbench now does exactly what our previous one did, but it’s much easier to create new tests: we only need to add lines to the `test_seq` sequence.

⁶to execute multiple statements in the same cycle, place them in an `action ... endaction` block.

There is another way we can simplify adding new tests. Most of our tests will have a stage checking that the counter is set to a specific value—which we achieved above with

```
if (counter.read() != 42) $display("FAIL: counter != 42");
```

Since all of this (other than the value 42) will remain the same in each test, we can save ourselves some typing by defining, in the module body, a function which takes the expected value as an argument and reports a failure when the counter does not match it:

```
function check(expected_val);
  action
    if (counter.read() != expected_val)
      $display("FAIL: counter != %0d", expected_val);
  endaction
endfunction
```

We can then rewrite `test_seq` in terms of `check`:

```
Stmt test_seq = seq
  counter.load(42);
  check(42);
  $display("TESTS FINISHED");
endseq;
```

and revel in the ease of adding tests.

8 Generalized Counter

There is no good reason why our counter couldn't work with any bit width: save for the number of wires, the arithmetic is the same. To achieve this, we should only need to replace all occurrences of 8 in the program with a type variable, and everything should work.

First, the interface: we'll make all of the methods parameterized on the bit-width, a type variable we'll call `size_t`:

```
interface Counter#(type size_t);
  method Bit#(size_t) read();
  method Action increment();
  method Action decrement();
  method Action load(Bit#(size_t) newval);
endinterface
```

Note that it's really the *interface*, not the *methods*, that is parameterized, so `size_t` remains the same in all of the methods.

We'll also need to change the type of the interface implemented by `mkCounter()`. Changing the module header to

```
module mkCounter(Counter#(8));
```

would work, but would again create an eight-bit counter; instead, we choose to parameterize the module as well:

```
module mkCounter(Counter#(size_t));
```

(from the lowercase letter starting `size_t`, `bsc` knows that it's a type variable, not a concrete type). `size_t` will be replaced with a concrete size when `mkCounter()` is instantiated.

Inside the module, the type of the register contents changes,

```
Reg#(Bit#(size_t)) value <- mkReg(0);
```

and so does the return type of `read()`,

```
method Bit#(size_t) read();
```

and the argument type of `load()`,

```
method Action load(Bit#(size_t) newval);
```

Finally, in our testbench, we must add the size parameter to the `mkCounter()` instantiation:

```
Counter#(8) counter <- mkCounter();
```

And that's all: `bsc` will fix `size_t` to 8 and automatically adjust related operations and assignments to the correct bit width.

Compiling our new counter, however, with

```
bsc -u -verilog TbCounter.bsv
```

results in an error:

```
Error: "MyCounter.bsv", line 10, column 8: (T0043)
  Cannot synthesize 'mkCounter': Its interface is polymorphic.
```

Why? It turns out that `bsc` does not produce parameterized RTL or simulators; because BSV parameterization mechanisms are more powerful than those of the RTL, this is often not even possible. So, instead of generating a separate Verilog module for `mkCounter.v`, we'll remove the `(* synthesize *)` attribute on `mkCounter()` and let `bsc` inline it in the design which instantiates it. Now the compilation works and the circuit simulates as previously.

9 Advanced Generalization

(Caveat: this is an advanced topic!)

But, really, the counter could work for any type... well, *almost*: any type on which addition is defined, and which can be converted to bits (to be stored in a register).⁷ This kind of constraint is expressed in BSV using a *proviso*: intuitively, a module can have a certain interface type *provided* that some requirement is satisfied.

Since the interface says nothing about how the implementation behind it should work, no provisos are necessary, and we just replace `Bit#(size_t)` with a type variable, say, `count_t`:

```
interface Counter#(type count_t);
  method count_t read();
  method Action increment();
  method Action decrement();
  method Action load(count_t newval);
endinterface
```

The module, however, requires a proviso, because our implementation requires addition and conversion to bits. The former is expressed as membership in the `Arith` typeclass, and the latter, as membership in the `Bits` typeclass:

```
module mkCounter(Counter#(count_t))
  provisos(Arith#(count_t), Bits#(count_t, count_t_sz));
```

(the `count_t_sz` is computed from `count_t` by `bsc` and available inside the module). We continue the replacement with `value`,

```
Reg#(count_t) value <- mkReg(0);
```

the `read()` method header,

```
method count_t read();
```

the `load()` method header,

```
method Action load(count_t newval);
```

and, finally, the counter instance in the testbench:

```
Counter#(Bit#(8)) counter <- mkCounter();
```

⁷values of type `Bool`, for example, cannot be added, and interfaces—such as `Counter`—cannot be converted to a fixed number of bits.

which propagates `Bit#(8)` to the relevant uses of `count_t`.

Now, after compilation, we have the same eight-bit counter as before. So what have we gained? We can now vary more than just the bit-width: we can easily change our design to count, say, twenty-bit signed integers,

```
Counter#(Int#(20)) counter <- mkCounter();
```

or any user-defined type for which addition is defined (for example, a complex number type or a floating-point number type).⁸

10 Summary

We started by developing a simple eight-bit counter with a simple testbench. Next, we learned how to compile and simulate BSV designs, and how to examine the waveforms. We then extended the behavior of the counter by adding a `decrement()` method—while trivially keeping all previous behavior and without having to alter caller modules (such as our testbench)—or other rules. Later, we made an improved—more extendible and maintainable—testbench using the FSM library and abstracting behavior in a function, and parameterized the counter design to any bit length and then to any possible type. While these examples have, by necessity, employed only some of the power of BSV, we hope they have succeeded as a brief introduction.

11 Exercises

Exercise 1 Add tests for `increment()` and `decrement()` to both versions of the testbench.

Exercise 2 In the `increment()`-only version of the counter, change `increment()` to take as an argument the amount by which to increment the counter.

Exercise 3 Write a “cycle” counter which only has `read()` and `load()` methods, and increments automatically on every cycle during which `load()` is not called.

Exercise 4 In the counter with simultaneous `increment()` and `decrement()`, change `increment()` and `decrement()` to take the change amount as an argument, as above. You might find either `mkWire()` or `mkRWire()` useful (both are defined in Prelude and always available).

⁸see the *Reference Guide* for information on defining addition on custom types.

Appendix A Chapter 3 code

```
// simple counter from Chapter 2

interface Counter;
  method Bit#(8) read();
  method Action load(Bit#(8) newval);
  method Action increment();
endinterface

(* synthesize *)
module mkCounter(Counter);
  Reg#(Bit#(8)) value <- mkReg(0);

  method Bit#(8) read();
    return value;
  endmethod

  method Action load(Bit#(8) newval);
    value <= newval;
  endmethod

  method Action increment();
    value <= value + 1;
  endmethod
endmodule
```

Appendix B Chapter 4 code

```
import Ch02_Counter::*;

(* synthesize *)
module mkTbCounter();
  Counter counter <- mkCounter();
  Reg#(Bit#(16)) state <- mkReg(0);

  rule step0(state == 0);
    counter.load(42);
    state <= 1;
  endrule

  rule step1(state == 1);
    if (counter.read() != 42) $display("FAIL: counter.load(42)");
  endrule
endmodule
```

```

        state <= 2;
    endrule

    rule done(state == 2);
        $display("TESTS FINISHED");
        $finish(0);
    endrule
endmodule

```

Appendix C Chapter 6 code

```

// counter + decrement from Chapter 5

interface Counter;
    method Bit#(8) read();
    method Action load(Bit#(8) newval);
    method Action increment();
    method Action decrement();
endinterface

(* synthesize *)
module mkCounter(Counter);
    Reg#(Bit#(8)) value <- mkReg(0);

    PulseWire increment_called <- mkPulseWire();
    PulseWire decrement_called <- mkPulseWire();

    rule do_increment(increment_called && !decrement_called);
        value <= value + 1;
    endrule

    rule do_decrement(!increment_called && decrement_called);
        value <= value - 1;
    endrule

    method Bit#(8) read();
        return value;
    endmethod

    method Action load(Bit#(8) newval);
        value <= newval;
    endmethod

```

```

method Action increment();
    increment_called.send();
endmethod

method Action decrement();
    decrement_called.send();
endmethod
endmodule

```

Appendix D Chapter 7 code

```

import StmtFSM::*;
import Ch05_Counter::*;

(* synthesize *)
module mkTbCounter();
    Counter counter <- mkCounter();
    Reg#(Bit#(16)) state <- mkReg(0);

    // check that the counter matches an expected value
    function check(expected_val);
        action
            if (counter.read() != expected_val)
                $display("FAIL: counter != %0d", expected_val);
        endaction
    endfunction

    Stmt test_seq = seq
        counter.load(42);
        check(42);
        $display("TESTS FINISHED");
    endseq;
    mkAutoFSM(test_seq);
endmodule

```

Appendix E Chapter 8 code

```

// generalized counter from Chapter 7

interface Counter#(type size_t);
    method Bit#(size_t) read();
    method Action load(Bit#(size_t) newval);
endinterface

```

```

    method Action increment();
    method Action decrement();
endinterface

module mkCounter(Counter#(size_t));
    Reg#(Bit#(size_t)) value <- mkReg(0);

    PulseWire increment_called <- mkPulseWire();
    PulseWire decrement_called <- mkPulseWire();

    rule do_increment(increment_called && !decrement_called);
        value <= value + 1;
    endrule

    rule do_decrement(!increment_called && decrement_called);
        value <= value - 1;
    endrule

    method Bit#(size_t) read();
        return value;
    endmethod

    method Action load(Bit#(size_t) newval);
        value <= newval;
    endmethod

    method Action increment();
        increment_called.send();
    endmethod

    method Action decrement();
        decrement_called.send();
    endmethod
endmodule

```

```

import StmtFSM::*;
import Ch07_Counter::*;

(* synthesize *)
module mkTbCounter();
    Counter#(8) counter <- mkCounter();

```

```

Reg#(Bit#(16)) state <- mkReg(0);

// check that the counter matches an expected value
function check(expected_val);
  action
    if (counter.read() != expected_val)
      $display("FAIL: counter != %0d", expected_val);
    endaction
endfunction

Stmt test_seq = seq
  counter.load(42);
  check(42);
  $display("TESTS FINISHED");
endseq;
mkAutoFSM(test_seq);
endmodule

```

Appendix F Chapter 9 code

```

// fully generalized counter from Chapter 8

interface Counter#(type count_t);
  method count_t read();
  method Action load(count_t newval);
  method Action increment();
  method Action decrement();
endinterface

module mkCounter(Counter#(count_t))
  provisos(Arith#(count_t), Bits#(count_t, count_t_sz));
  Reg#(count_t) value <- mkReg(0);

  PulseWire increment_called <- mkPulseWire();
  PulseWire decrement_called <- mkPulseWire();

  rule do_increment(increment_called && !decrement_called);
    value <= value + 1;
  endrule

  rule do_decrement(!increment_called && decrement_called);
    value <= value - 1;
  endrule
endmodule

```

```

method count_t read();
    return value;
endmethod

method Action load(count_t newval);
    value <= newval;
endmethod

method Action increment();
    increment_called.send();
endmethod

method Action decrement();
    decrement_called.send();
endmethod
endmodule

import StmtFSM::*;
import Ch08_Counter::*;

(* synthesize *)
module mkTbCounter();
    Counter#(Bit#(8)) counter <- mkCounter();
    Reg#(Bit#(16)) state <- mkReg(0);

    // check that the counter matches an expected value
    function check(expected_val);
        action
            if (counter.read() != expected_val)
                $display("FAIL: counter != %0d", expected_val);
        endaction
    endfunction

    Stmt test_seq = seq
        counter.load(42);
        check(42);
        $display("TESTS FINISHED");
    endseq;
    mkAutoFSM(test_seq);
endmodule

```